<commit_before>{ stdenv, fetchFromGitHub, gcc, jdk, which }:

stdenv.mkDerivation rec {
  name = "${pname}-${version}";
  pname = "drip";
  version = "0.2.4";

  src = fetchFromGitHub {
    repo = pname;
    owner = "ninjudd";
    rev = version;
    sha256 = "1zl62wdwfak6z725asq5lcqb506la1aavj7ag78lvp155wyh8aq1";
  };

  buildInputs = [ gcc jdk which ];
 
  patchPhase = ''
    mkdir $out
    cp ./* $out -r
  '';

  buildPhase = ''
    $out/bin/drip version
  '';

  phases = [ "unpackPhase" "patchPhase" "buildPhase" ];

  meta = with stdenv.lib; {
    description = "a launcher for the Java Virtual Machine intended to be a drop-in replacement for the java command, only faster";
    license = licenses.epl10;
    homepage = https://github.com/ninjudd/drip;
    platforms = platforms.linux;
    maintainers = [ maintainers.rybern ];
  };
}
<commit_msg>drip: provided which at run time<commit_after>{ stdenv, fetchFromGitHub, jdk, which, makeWrapper }:

stdenv.mkDerivation rec {
  name = "${pname}-${version}";
  pname = "drip";
  version = "0.2.4";

  src = fetchFromGitHub {
    repo = pname;
    owner = "ninjudd";
    rev = version;
    sha256 = "1zl62wdwfak6z725asq5lcqb506la1aavj7ag78lvp155wyh8aq1";
  };

  nativeBuildInputs = [ makeWrapper ];

  buildInputs = [ jdk ];

  postPatch = ''
    patchShebangs .
  '';

  installPhase = ''
    runHook preInstall
    mkdir $out
    cp ./* $out -r
    wrapProgram $out/bin/drip \
      --prefix PATH : "${which}/bin"
    $out/bin/drip version
    runHook postInstall
  '';

  meta = with stdenv.lib; {
    description = "A launcher for the Java Virtual Machine intended to be a drop-in replacement for the java command, only faster";
    license = licenses.epl10;
    homepage = https://github.com/ninjudd/drip;
    platforms = platforms.linux;
    maintainers = [ maintainers.rybern ];
  };
}

{'fraction_non_alphanumeric': 0.13027027027027027, 'fraction_numerical': 0.03027027027027027, 'mean_word_length': 3.015184381778742, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4612404', 'n_tokens_mistral': 714, 'n_tokens_neox': 673, 'n_words': 155}